

ParBoids

Catelen Wu
cw3223@columbia.edu

Ethan Wu
ew2664@columbia.edu

1 Introduction

Boids (“bird-oids”) is an artificial life program simulating the flocking behavior of birds developed by Craig Reynolds in 1986. It is an example of emergent behavior and swarm intelligence: each boid agent follows only a simple set of rules, but the interactions between them give rise to complex and unpredictable behavior mimicking that of flocks or herds of animals found in nature. In this project, we first develop a sequential **Boids** simulation in Haskell, then attempted to parallelize the program using different strategies.




Figure 1: Steering forces of separation, alignment, and cohesion (left to right)

In **Boids**, each boid has its individual position, velocity, and mass. Each boid is also subject to three steering forces: **separation**, **alignment**, and **cohesion**. (Other rules can also be added to simulate more complex behavior, such as follow-the-leader or obstacle avoidance.) These forces are computed based on the relative positions and velocities of each boid to its local flockmates. Other boids beyond a certain radius are ignored. Then, every time step, the position and velocity of each boid is updated by a weighted sum of the three steering forces. These updates are repeated indefinitely until the program is terminated. The main update loop, run for n time steps on flock \mathcal{B} , is shown in Algorithm 1 below.

Algorithm 1 Reynolds’s Flocking Algorithm

```
for  $i \leftarrow 1, n$  do
  for each  $b \in \mathcal{B}$  do
    nbs  $\leftarrow$  neighbors( $b, \mathcal{B}$ )                                 $\triangleright$  Get local flockmates
     $f_s \leftarrow$  separation( $b, \text{nbs}$ )
     $f_a \leftarrow$  alignment( $b, \text{nbs}$ )
     $f_c \leftarrow$  cohesion( $b, \text{nbs}$ )
     $b_v \leftarrow b_v + (k_s f_s + k_a f_a + k_c f_c) \Delta t / b_m$      $\triangleright$  Update boid velocity
     $b_x \leftarrow b_x + b_v \Delta t$                                  $\triangleright$  Update boid position
  end for
end for
```

2 Sequential Implementation

A sequential implementation for Reynolds’s flocking simulation is relatively simple. First, we define a **Boid** data type using Haskell’s record syntax, as well as some utility functions for computing the displacement

between two boids and finding a boid's local flockmates. We use `Linear.V2` from the `linear` package to represent each boid's position and velocity vectors in two dimensions.

```
data Boid = Boid { bPos :: V2 Float, bVel :: V2 Float, bMass :: Float } deriving (Show)

between :: Config -> Boid -> Boid -> V2 Float
between cfg b bo = case wSize cfg of
  Size size -> wrapDisp size (bPos b) (bPos bo)
  Infinite -> bPos bo ^-^ bPos b

flockmates :: Config -> [Boid] -> Float -> Boid -> [(Boid, V2 Float)]
flockmates cfg flock r b = filter (\(bo, _) -> bPos bo /= bPos b) neighbors
  where
    neighbors = takeWhile (\(_, disp) -> norm disp < r) sorted
    sorted = sortOn (norm . snd) $ map (\bo -> (bo, between cfg b bo)) flock
```

We also define separate functions for each of the three steering forces that computes the effect on a boid from a neighbor. Each force is similarly represented with a two-dimensional `V2` vector.

```
separation :: Boid -> (Boid, V2 Float) -> V2 Float
separation _ (_, disp) = negated disp ^/ (norm disp ** 2)

alignment :: Boid -> (Boid, V2 Float) -> V2 Float
alignment b (bo, _) = bVel bo ^-^ bVel b

cohesion :: Boid -> (Boid, V2 Float) -> V2 Float
cohesion _ (_, disp) = disp
```

Next, we define a `Steer` data type to collect the steering forces acting on a boid.

```
data Steer = Steer { sSf :: V2 Float, sAf :: V2 Float, sCf :: V2 Float } deriving (Show)

initSteer :: Steer
initSteer = Steer zero zero zero
```

Then, for each boid, we have a function `updateBoid` that iterates through its local flockmates, accumulates the steering forces in a `Steer` record, and update's the boid's position and velocity with the net force.

```
steerFrom :: Boid -> Steer -> (Boid, V2 Float) -> Steer
steerFrom b (Steer sf af cf) disp = Steer sf' af' cf'
  where
    sf' = sf ^+^ separation b disp
    af' = af ^+^ alignment b disp
    cf' = cf ^+^ cohesion b disp

updateBoid :: Config -> [Boid] -> Boid -> Boid
updateBoid cfg flock b = b {bPos = pos', bVel = vel'}
  where
    pos' = wrapPos cfg $ bPos b ^+^ 0.1 *^ vel'
    vel' = vBound (maxVel cfg) (bVel b ^+^ 0.05 *^ netf ^/ bMass b)
    netf = sn cfg *^ sf ^+^ an cfg *^ af ^+^ cn cfg *^ cf
    Steer sf af cf = foldl (steerFrom b) initSteer bs
```

```
bs = flockmates cfg flock (radius cfg) b
```

Finally, we have our main simulation loop `runSimCollect` that runs recursively for `nIter` iterations, updating each boid in the flock using `updateBoid` in each iteration. The state of the flock after each iteration is collected in order to be written to file output or rendered in animation. In the alternative function `runSim`, only the last flock state is kept and all its predecessors are discarded.

```
runSim :: Config -> [Boid] -> Int -> [Boid]
runSim config flock0 nIter = case runSimCollect config flock0 nIter of
  [] -> flock0
  (flockN : _) -> flockN

runSimCollect :: Config -> [Boid] -> Int -> [[Boid]]
runSimCollect cfg flock0 nIter = foldl simLoop [flock0] [1 .. nIter]
  where
    simLoop :: [[Boid]] -> Int -> [[Boid]]
    simLoop [] _ = []
    simLoop flocks@(flock : _) _ = map (updateBoid cfg flock) flock : flocks
```

We also wrote a supplementary animation function using the `gloss` library to visualize the results of the simulation. Figure 2 below shows a frame from our animation of a flock of 50 boids.




Figure 2: Example frame from an animation of a flock of 50 boids

3 Parallel Implementation

Reynolds's flocking algorithm actually lends itself naturally to parallelization because the update at each time step is only dependent upon the state of the flock at the previous time step. That is, given the previous state of the flock, each boid's update is independent. Therefore, our main approach to parallelizing our sequential implementation is to perform boid updates in parallel.

First, we refactored our `runSimCollect` function to allow us to easily plug in different update methods implemented with different strategies. We define a new data type `ParStrat` to indicate the strategy being used. The sequential `map` over boids from above is moved into the `updateSeq` function.

```
data ParStrat = Seq | TwoPart | Chunks Int | ParList

updateWith :: ParStrat -> Config -> [Boid] -> [Boid]
updateWith Seq = updateSeq
```

```

updateWith TwoPart = updateTwoPart
updateWith (Chunks n) = updateChunks n
updateWith ParList = updateParList

runSimCollect :: Config -> [Boid] -> Int -> [[Boid]]
runSimCollect cfg flock0 nIter = foldl simLoop [flock0] [1 .. nIter]
  where
    simLoop :: [[Boid]] -> Int -> [[Boid]]
    simLoop [] _ = []
    simLoop flocks@(flock : _) _ = updateWith Seq cfg flock : flocks

updateSeq :: Config -> [Boid] -> [Boid]
updateSeq cfg flock = map (updateBoid cfg flock) flock

```

The first strategy we attempted, `TwoPart`, is static two-way partitioning, where we split the flock into two sub-flocks and update each flock in parallel using `rpar`. Even though the work needed for each boid may differ depending on its number of neighbors, we don't expect this difference to be too great given the small radius of each boid's neighborhood. Moreover, with a sufficient number of boids, the work needed to update the two sub-flocks will even out, so we believed this strategy to be a reasonable initial approach.

```

updateTwoPart :: Config -> [Boid] -> [Boid]
updateTwoPart cfg flock = runEval $ do
  as' <- rpar (force (map (updateBoid cfg flock) as))
  bs' <- rpar (force (map (updateBoid cfg flock) bs))
  _ <- rseq as'
  _ <- rseq bs'
  return (as' ++ bs')
  where
    (as, bs) = splitAt (length flock `div` 2) flock

```

A more sophisticated version of the above approach that we attempted next is `Chunks`. This approach uses `parListChunk` from `Control.Parallel.Strategies` to split the flock into a specified number of sub-flocks, spark an update for each sub-flock, and recombine the result.

```

updateChunks :: Int -> Config -> [Boid] -> [Boid]
updateChunks numChunks cfg flock = flock'
  where
    flock' = map (updateBoid cfg flock) flock `using` parListChunk numChunks rdeepseq

```

The final approach we attempted is `ParList`, using `parList` from `Control.Parallel.Strategies`. This sparks an update for each individual boid, equivalent to `Chunks` with a chunk size of 1.

```

updateParList :: Config -> [Boid] -> [Boid]
updateParList cfg flock = flock'
  where
    flock' = map (updateBoid cfg flock) flock `using` parList rdeepseq

```

Note that in the above approaches, for `force` and `rdeepseq` from `Control.DeepSeq` to be able to fully evaluate each `Boid` to normal form, we define a rather trivial `NFData` instance for `Boid`.

```

instance NFData Boid where

```

```
rnf (Boid pos vel m) = rnf pos `seq` rnf vel `seq` rnf m
```

We discuss the results of our experimentation on these various approaches in the next section.

4 Results and Discussion

After implementing our various strategies as described above, we experimented with different parameters in order to gauge the effectiveness of each strategy. First, we experimented with varying the size of our flock, ranging from 50 boids to 5000 boids. Specifically, we measured the total time it took for the baseline sequential algorithm to simulate 100 iterations for these various flock sizes. We then compared this to the performance of static two-way partitioning running on two cores. The execution times are plotted in Figure 3 below, with example Threadscope profiles of the two strategies shown in Figure 4 and Figure 5.




Figure 3: Results from Seq and TwoPart on different numbers of boids




Figure 4: Seq Threadscope profile for 500 boids




Figure 5: TwoPart Threadscope profile for 500 boids

First, we observe that regardless of parallelization, there appears to be a roughly $O(N^2)$ increase in execution time as the flock size increases. This is along the lines of what we expect, especially for large flock sizes, because the bottleneck in the algorithm becomes finding each boid’s closest flock-mates. (Our implementation just uses a linear filter, though there may be more efficient solutions such as storing the boids in a quadtree structure.) We also see that two-way partitioning parallelizes work decently with both cores being utilized evenly. As we hoped, we did not encounter the problem of unbalanced partitions because work for each partition tends to even out as flock size increases. However, we do see that a big portion of time in both cores is taken up by garbage collection, causing activity to be spiky and significantly below the full potential.

Next, we moved on to testing our two other approaches `Chunks` and `ParList`, which use the `parListChunk` and `parList` evaluation strategies respectively. These two strategies can take advantage of more cores to hopefully provide further speedups. For each trial, we ran our parallel algorithm on 500 boids for 100 iterations. We experimented with both strategies using different numbers of cores, from 1 and up to 8. For the former, we also varied the number of chunks we used. The execution times are plotted in Figure 6 below, with example Threadscope profiles of the two strategies shown in Figure 7 and Figure 8.




Figure 6: Results from `Chunks` and `ParList` using different numbers of cores and chunks




Figure 7: `Chunks 50` Threadscope profile for 500 boids

We can make a few interesting observations from Figure 6. First, all strategies generally decrease execution time as the number of cores increase. (An exception is `Chunks 5`, which as can be expected, stop receiving gains after more than 5 cores were used.) For the other strategies, there were also diminishing returns, often with a number of cores beyond which adding more cores is no longer produces a speed-up. For example, for `Chunks 50`, which split the flock into 10 chunks, using 7 cores was optimal; at 8 cores, the total execution time increased again. These diminishing returns are partly due to Amdahl’s law, as our algorithm deals with a significant amount of IO (e.g., reading and saving the state of the flock to file) that is inherently sequential. Furthermore, there is increased overhead and garbage collection: we found that as the number of cores increases, the amount of garbage collection increases noticeably, as seen in the Threadscope profiles.




Figure 8: ParList Threadscape profile for 500 boids

In a similar regard, for `Chunks`, increasing the number of chunks initially produced better performance, up to a point where more chunks results in too many sparks, more overhead, and poorer performance. Where this point is may depend on the number of cores. For example, we see that `ParList` (equivalent to `Chunks` 500 for our case of 500 boids) was initially one of the worst performers, but improved relative to the other strategies when we used higher numbers of cores that could more efficiently process the sparks it generated. Overall, we found the optimal number of chunks to be somewhere around 50, which produced consistently lower execution times for all numbers of cores; the corresponding optimal number of cores is around 7.

5 Code

5.1 app/Main.hs

```

1  module Main (main) where
2
3  import Animate (runAnimation)
4  import BoidIO (loadFlock, saveFlock)
5  import Config (loadConfig)
6  import Control.Monad (foldM_, unless)
7  import GHC.Base (when)
8  import Options.Applicative
9  import Sim (runSimCollect)
10 import System.Exit (die)
11
12 data Args = Arguments
13   { flockFile :: String,
14     numIter :: Int,
15     outputDir :: Maybe String,
16     configFile :: Maybe String,
17     animate :: Bool
18   }
19
20 arguments :: Parser Args
21 arguments =
22   Arguments
23     <$> argument str (metavar "FILE" <> help "Initial flock data file")
24     <*> option auto (long "num-iter" <> short 'n' <> metavar "INT" <> help "Number of iterations")
25     <*> optional (strOption (long "out-dir" <> short 'o' <> metavar "DIR" <> help "Output directory"))
26     <*> optional (strOption (long "config" <> short 'c' <> metavar "CONFIG" <> help "Configuration file"))
27     <*> switch (long "animate" <> short 'a' <> help "Whether to run animation")
28
29 main :: IO ()
30 main = run =<< execParser opts

```

```

31  where
32      opts =
33          info
34              (arguments <*> helper)
35              ( fullDesc
36                  <> progDesc ""
37                  <> header ""
38              )
39
40  run :: Args -> IO ()
41  run args = do
42      unless (numIter args > 0) $ die "num-iter must be a positive integer"
43      flock0 <- loadFlock $ flockFile args
44      config <- loadConfig $ configFile args
45      print config
46      let flocks = reverse $ runSimCollect config flock0 $ numIter args
47      foldM_ (saveFlock $ outputDir args) 0 flocks
48      putStrLn "simulation complete"
49      when (animate args) $ do
50          putStrLn "running animation"
51          runAnimation flocks
52      putStrLn "process complete"

```

5.2 src/Animate.hs

```

1  module Animate (runAnimation) where
2
3  import Boid (Boid, bPos, bVel)
4  import Graphics.Gloss
5  import Graphics.Gloss.Data.ViewPort (ViewPort)
6  import Linear.Vector ((*^), (^+^))
7  import Utils (vScaleTo, vxy)
8
9  background :: Color
10 background = white
11
12 window :: Display
13 window = InWindow "ParBoids" (800, 600) (200, 200)
14
15 update :: ViewPort -> Float -> [[Boid]] -> [[Boid]]
16 update _ _ [] = []
17 update _ _ (_ : flocks) = flocks
18
19 render :: [[Boid]] -> Picture
20 render [] = blank
21 render (flock : _) = pictures $ map draw flock
22
23 draw :: Boid -> Picture
24 draw boid =
25     pictures
26     [ translate x y $ color red $ circleSolid 3,
27       translate x' y' $ color blue $ circleSolid 2
28     ]
29     where
30         (x', y') = vxy $ scaleFac *^ bPos boid ^+^ vScaleTo 2 (bVel boid)
31         (x, y) = vxy $ scaleFac *^ bPos boid

```



```

32     scaleFac = 20
33
34     runAnimation :: [[Boid]] -> IO ()
35     runAnimation flocks = simulate window background 60 flocks render update

```

5.3 src/Boid.hs

```

1     module Boid (Boid, newBoid, updateBoid, bPos, bVel) where
2
3     import Config (Config (..), WorldSize (..))
4     import Control.DeepSeq (NFData (..))
5     import Data.List (sortOn)
6     import Linear (negated)
7     import Linear.Metric (Metric (norm))
8     import Linear.V2 (V2 (..))
9     import Linear.Vector (zero, (*^), (^+^), (^-^), (^/))
10    import Utils (vBound, vWrap, wrapDisp)
11
12    data Boid = Boid { bPos :: V2 Float, bVel :: V2 Float, bMass :: Float } deriving (Show)
13
14    instance NFData Boid where
15        rnf (Boid pos vel m) = rnf pos `seq` rnf vel `seq` rnf m
16
17    newBoid :: [Float] -> Maybe Boid
18    newBoid [px, py, vx, vy, m] = Just $ Boid (V2 px py) (V2 vx vy) m
19    newBoid [px, py, vx, vy] = Just $ Boid (V2 px py) (V2 vx vy) 1
20    newBoid _ = Nothing
21
22    between :: Config -> Boid -> Boid -> V2 Float
23    between cfg b bo = case wSize cfg of
24        Size size -> wrapDisp size (bPos b) (bPos bo)
25        Infinite -> bPos bo ^-^ bPos b
26
27    flockmates :: Config -> [Boid] -> Float -> Boid -> [(Boid, V2 Float)]
28    flockmates cfg flock r b = filter (\(bo, _) -> bPos bo /= bPos b) neighbors
29        where
30            neighbors = takeWhile (\(_, disp) -> norm disp < r) sorted
31            sorted = sortOn (norm . snd) $ map (\bo -> (bo, between cfg b bo)) flock
32
33    wrapPos :: Config -> V2 Float -> V2 Float
34    wrapPos cfg pos = case wSize cfg of
35        Size size -> vWrap size pos
36        Infinite -> pos
37
38    data Steer = Steer { sSf :: V2 Float, sAf :: V2 Float, sCf :: V2 Float } deriving (Show)
39
40    initSteer :: Steer
41    initSteer = Steer zero zero zero
42
43    steerFrom :: Boid -> Steer -> (Boid, V2 Float) -> Steer
44    steerFrom b (Steer sf af cf) disp = Steer sf' af' cf'
45        where
46            sf' = sf ^+^ separation b disp
47            af' = af ^+^ alignment b disp
48            cf' = cf ^+^ cohesion b disp
49

```

```

50 updateBoid :: Config -> [Boid] -> Boid -> Boid
51 updateBoid cfg flock b = b {bPos = pos', bVel = vel'}
52   where
53     pos' = wrapPos cfg $ bPos b ^+^ 0.1 *^ vel'
54     vel' = vBound (maxVel cfg) (bVel b ^+^ 0.05 *^ netf ^/ bMass b)
55     netf = sn cfg *^ sf ^+^ an cfg *^ af ^+^ cn cfg *^ cf
56     Steer sf af cf = foldl (steerFrom b) initSteer bs
57     bs = flockmates cfg flock (radius cfg) b
58
59 separation :: Boid -> (Boid, V2 Float) -> V2 Float
60 separation _ (_, disp) = negated disp ^/ (norm disp ** 2)
61
62 alignment :: Boid -> (Boid, V2 Float) -> V2 Float
63 alignment b (bo, _) = bVel bo ^-^ bVel b
64
65 cohesion :: Boid -> (Boid, V2 Float) -> V2 Float
66 cohesion _ (_, disp) = disp

```

5.4 src/BoidIO.hs

```

1  module BoidIO (loadFlock, saveFlock) where
2
3  import Boid (Boid, newBoid)
4  import System.IO (Handle, IOMode (ReadMode, WriteMode), hClose, hGetLine, hIsEOF, hPrint, withFile)
5
6  loadFlock :: String -> IO [Boid]
7  loadFlock file = withFile file ReadMode readFlockFile
8
9  readFlockFile :: Handle -> IO [Boid]
10 readFlockFile hdl = do
11   isEOF <- hIsEOF hdl
12   ( if isEOF
13     then return []
14     else
15       ( do
16         line <- hGetLine hdl
17         bs <- readFlockFile hdl
18         case newBoid $ map read (words line) of
19           Just b -> return (b : bs)
20           Nothing -> return bs
21       )
22   )
23
24 saveFlock :: Maybe String -> Int -> [Boid] -> IO Int
25 saveFlock outDir n = case outDir of
26   Just dn -> \bs -> do
27     withFile file WriteMode $ writeFlockFile bs
28     return (n + 1)
29   where
30     file = dn ++ "/" ++ show n ++ ".txt"
31   Nothing -> \_ -> return 0
32
33 writeFlockFile :: [Boid] -> Handle -> IO ()
34 writeFlockFile [] hdl = hClose hdl
35 writeFlockFile (b : bs) hdl = do
36   hPrint hdl b

```

```
37 writeFlockFile bs hdl
```

5.5 src/Config.hs

```
1  module Config (Config (..), WorldSize (..), loadConfig) where
2
3  import System.IO (Handle, IOMode (ReadMode), hGetLine, hIsEOF, withFile)
4
5  data Config = Config
6    { radius :: Float,
7      sn     :: Float,
8      an     :: Float,
9      cn     :: Float,
10     maxVel :: Float,
11     wSize  :: WorldSize
12   }
13   deriving (Show)
14
15   data WorldSize = Infinite | Size Float
16
17   instance Show WorldSize where
18     show Infinite = ""
19     show (Size f) = show f
20
21   defaultConfig :: Config
22   defaultConfig =
23     Config
24     { radius = 5,
25       sn     = 1.8,
26       an     = 0.08,
27       cn     = 0.3,
28       maxVel = 10,
29       wSize  = Infinite
30     }
31
32   loadConfig :: Maybe String -> IO Config
33   loadConfig file = case file of
34     Just fn -> withFile fn ReadMode readConfigFile
35     Nothing -> return defaultConfig
36
37   readConfigFile :: Handle -> IO Config
38   readConfigFile hdl = do
39     isEOF <- hIsEOF hdl
40     ( if isEOF
41       then return defaultConfig
42       else
43         ( do
44           line <- hGetLine hdl
45           cfg <- readConfigFile hdl
46           let cfg' = case words line of
47                 ["radius", arg] -> cfg {radius = read arg}
48                 ["sn", arg]     -> cfg {sn = read arg}
49                 ["an", arg]     -> cfg {an = read arg}
50                 ["cn", arg]     -> cfg {cn = read arg}
51                 ["maxVel", arg] -> cfg {maxVel = read arg}
```

```

52         ["wSize", arg] -> cfg {wSize = Size $ read arg}
53     _ -> cfg
54     return cfg'
55 )
56 )

```

5.6 src/Sim.hs

```

1  module Sim (runSim, runSimCollect) where
2
3  import Boid (Boid, updateBoid)
4  import Config (Config)
5  import Control.DeepSeq (force)
6  import Control.Parallel.Strategies (parList, parListChunk, rdeepseq, rpar, rseq, runEval, using)
7
8  data ParStrat = Seq | TwoPart | Chunks Int | ParList
9
10 updateWith :: ParStrat -> Config -> [Boid] -> [Boid]
11 updateWith Seq = updateSeq
12 updateWith TwoPart = updateTwoPart
13 updateWith (Chunks n) = updateChunks n
14 updateWith ParList = updateParList
15
16 runSim :: Config -> [Boid] -> Int -> [Boid]
17 runSim config flock0 nIter = case runSimCollect config flock0 nIter of
18     [] -> flock0
19     (flockN : _) -> flockN
20
21 runSimCollect :: Config -> [Boid] -> Int -> [[Boid]]
22 runSimCollect cfg flock0 nIter = foldl simLoop [flock0] [1 .. nIter]
23     where
24         simLoop :: [[Boid]] -> Int -> [[Boid]]
25         simLoop [] _ = []
26         simLoop flocks@(flock : _) _ = updateWith Seq cfg flock : flocks
27
28 updateSeq :: Config -> [Boid] -> [Boid]
29 updateSeq cfg flock = map (updateBoid cfg flock) flock
30
31 updateTwoPart :: Config -> [Boid] -> [Boid]
32 updateTwoPart cfg flock = runEval $ do
33     as' <- rpar (force (map (updateBoid cfg flock) as))
34     bs' <- rpar (force (map (updateBoid cfg flock) bs))
35     _ <- rseq as'
36     _ <- rseq bs'
37     return (as' ++ bs')
38     where
39         (as, bs) = splitAt (length flock `div` 2) flock
40
41 updateChunks :: Int -> Config -> [Boid] -> [Boid]
42 updateChunks numChunks cfg flock = flock'
43     where
44         flock' = map (updateBoid cfg flock) flock `using` parListChunk chunkSize rdeepseq
45         chunkSize = length flock' `div` numChunks
46
47 updateParList :: Config -> [Boid] -> [Boid]
48 updateParList cfg flock = flock'

```

```

49     where
50     flock' = map (updateBoid cfg flock) flock `using` parList rdeepseq

```

5.7 src/Utils.hs

```

1  module Utils (vBound, vScaleTo, vx, vy, vxy, wrapDisp, vWrap) where
2
3  import Data.Fixed (mod')
4  import Linear.Metric (Metric (norm), normalize)
5  import Linear.V2 (V2 (V2))
6  import Linear.Vector ((*^))
7
8  vBound :: Float -> V2 Float -> V2 Float
9  vBound lim v = vScaleTo (norm v `min` lim) v
10
11 vScaleTo :: Float -> V2 Float -> V2 Float
12 vScaleTo n v = n *^ normalize v
13
14 vWrap :: Float -> V2 Float -> V2 Float
15 vWrap size (V2 x y) = V2 (wrap x) (wrap y)
16     where
17     wrap a = (a + size / 2) `mod` size - (size / 2)
18
19 vx :: V2 a -> a
20 vx (V2 x _) = x
21
22 vy :: V2 a -> a
23 vy (V2 _ y) = y
24
25 vxy :: V2 a -> (a, a)
26 vxy (V2 x y) = (x, y)
27
28 wrapDisp :: Float -> V2 Float -> V2 Float -> V2 Float
29 wrapDisp size p1 p2 = V2 dx' dy'
30     where
31     dx'
32     | abs dx > 0.5 * size = dx + (if x2 > x1 then -size else size)
33     | otherwise = dx
34     dy'
35     | abs dy > 0.5 * size = dy + (if y2 > y1 then -size else size)
36     | otherwise = dy
37     dx = x2 - x1
38     dy = y2 - y1
39     (V2 x1 y1) = vWrap size p1
40     (V2 x2 y2) = vWrap size p2

```